# $Id$
# Maintainer: Jan de Groot <jgc@archlinux.org>

pkgname=xorg-smproxy
pkgver=1.0.6
pkgrel=2
pkgdesc="Allows X applications that do not support X11R6 session management to participate in an X11R6 session"
arch=(x86_64)
url="https://xorg.freedesktop.org/"
license=('custom')
depends=('libsm' 'libxt' 'libxmu')
makedepends=('xorg-util-macros')
groups=('xorg-apps' 'xorg')
source=(https://xorg.freedesktop.org/archive/individual/app/smproxy-${pkgver}.tar.bz2{,.sig})
sha512sums=('4945c92c0f07c5dba5698889f415e78ce406939f4a36eb977309488c27385ce4ce04235c4e850a90a451fea8d7885ded7c6b18db71335c7731b38aaeefa9fcb7'
            'SKIP')
validpgpkeys=('4A193C06D35E7C670FA4EF0BA2FB9E081F2D130E') # Alan Coopersmith <alan.coopersmith@oracle.com>

build() {
  cd smproxy-${pkgver}
  ./configure --prefix=/usr
  make
}

package() {
  cd smproxy-${pkgver}
  make DESTDIR="${pkgdir}" install
  install -m755 -d "${pkgdir}/usr/share/licenses/${pkgname}"
  install -m644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/"
}
